
String Functions

The most common operation with string values, addition (concatenation), was discussed in the Operations and Expressions (Problem 3) section. In some cases, there is a need in performing other calculations related to string values. MQL4 has a number of string functions for working with the values of string type. Let's consider the usage of some of them through the example below.

	
	Problem 35. Color the last 100 bars of the candlestick chart as follows: black candlesticks in red, white candlesticks in blue.


A candlestick can be colored using two lines: a thin line must overlay a candlestick so that it covers all the shadows, whereas a thick line should fill out a candlestick body. In this case, we cannot use the lines of a custom indicator, because the displayed lines must be vertical, i.e., constructed using two coordinates (with the same time coordinates), while indicator arrays allow us to store only one value set in correspondence with each bar. So, the problem solution comes to displaying a series of single-type OBJ_TREND objects that differ in their coordinates and line style and color (see Graphical Objects) on a price chart. 
In this case, the EA is used as an application program, but, in general, the algorithm can be implemented in a custom indicator. As a whole, the algorithm is clear. The chart must be colored for the first time, as soon as it is attached to the symbol window (during the execution of init()). The program must track possible changes in the location of graphical objects (user can accidentally move or delete one of them) with every tick coming, and restore them, if necessary. All objects created by the program must be deleted, as soon as the program finishes operating (deinit()).

A user can create other objects in a symbol window while the EA is working, for example, place the channel of standard deviations, Fibo levels, support lines, etc. So, the algorithm that allows us to distinguish user-created and program-created objects must be implemented in the program. This is particularly important when closing the program: it is necessary to remove only the program-created objects, while the user-created objects must remain unchanged. Each graphical object has its own properties that can generally coincide. The only identifying feature of any object is its unique name (the use of the same names is prohibited).

It's recommended to enter the useful information in the object's name while composing it, so it will be possible to detect the location and the properties of the object. For example, an object name may contain a prefix that differentiates a program-created object from others. In this case, it is "Paint_". Besides, it is necessary to differentiate the "user-defined" objects from any other, as well. The same time a simple numeration (Paint_1, Paint_2) cannot be used. Using this method of objects numeration, you cannot understand, at which bar the object Paint_73 should be displayed. The bar that has the Paint_73 index will get the Paint_74 index, when a new bar comes, Paint_75 index when another new bar comes, etc. In such a case, it would be necessary to delete and re-create all the objects on every new bar. This solution (although it is possible) is obviously very rough and expensive.

Every object created must have its time coordinates that correspond with the time of bar opening. Besides, two lines must be displayed on every bar - a thin line and a thick line. It is most comfortable to represent the names of a created objects by the program as follows:

Object name = Paint_2_2007.03.22 16:40, here:

Paint_ - prefix that differentiates the objects created by the program;

2_ - number of either objects that are displayed on a bar (value 1 or 2 is possible);

2007.03.22 16:40 - time coordinate that uniquely characterizes the bar the object is displayed on.

Paint_ and 2_ are the values of the variables Prefix and Nom_Lin, respectively. The time coordinate can be obtained for every bar by transformation a datetime value into a string value using the transforming functions:

TimeToStr() Function

string TimeToStr(datetime value, int mode=TIME_DATE|TIME_MINUTES)
The function transforms the values that contain time (in seconds) lapsed since 01.01.1970 (datetime value) into a string of the specified format (string value).

Parameters:

value - time in seconds lapsed since 00:00 of the 1st of January 1970;

mode - an additional mode of data output. It can be a single or a combined flag:

TIME_DATE obtains the result in the "yyyy.mm.dd" form;

TIME_MINUTES obtains the result in the "hh:mi" form;

TIME_SECONDS obtains the result in the "hh:mi:ss" form.

Let's consider the EA strings.mq4 that manages objects for coloring of candles and see how the TineToStr() is used in this program:

//--------------------------------------------------------------------
// strings.mq4
// The code should be used for educational purpose only.
//--------------------------------------------------------------- 1 --
extern int Quant_Bars=100;             // Number of bars
datetime   Time_On;
string     Prefix    ="Paint_";
//--------------------------------------------------------------- 2 --
int init()                             // Spec. function init()
  {
   int Ind_Bar;                        // Bar index
   Time_On=Time [Quant_Bars];          // Time of first coloring
   for(Ind_Bar=Quant_Bars-1; Ind_Bar>=0; Ind_Bar--)// Bars cycle
     {
      Create(Ind_Bar,1);               // Draw a thin line
      Create(Ind_Bar,2);               // Draw a thick line
     }
   WindowRedraw();                     // Image redrawing
   return;                             // Exit init()
  }
//--------------------------------------------------------------- 3 --
int start()                            // Spec. function start()
  {
   datetime T1, T2;                    // 1 and 2 time coordinates
   int Error,Ind_Bar;                  // Error code and bar index
   double P1, P2;                      // 1 and 2 price coordinates
   color Col;                          // Color of created object
//--------------------------------------------------------------- 4 --
   for(int Line=1; Line<=2; Line++)    // Line type cycle
     {
      string Nom_Lin =Line + "_";      // String with the line number
      //    string Nom_Lin  = DoubleToStr(Line,0)+"_";// Can be so
      for(Ind_Bar=0; ;Ind_Bar++)       // Bar cycle
        {
//--------------------------------------------------------------- 5 --
         datetime T_Bar= Time[Ind_Bar];// Bar opening time
         if (T_Bar < Time_On) break;   // Don't color out of borders
         string Str_Time=TimeToStr(T_Bar);       // Time string
         string His_Name=Prefix+Nom_Lin+Str_Time;// Object name
//--------------------------------------------------------------- 6 --
         T1=ObjectGet(His_Name,OBJPROP_TIME1);// t1 coord. query
         Error=GetLastError();         // Error code receiving
         if (Error==4202)              // If there is no object :(
           {
            Create(Ind_Bar,Line);      // Object creating function call.
            continue;                  // To the next iteration
           }
//--------------------------------------------------------------- 7 --
         T2 =ObjectGet(His_Name,OBJPROP_TIME2); // t2 coord. query
         P1 =ObjectGet(His_Name,OBJPROP_PRICE1);// p1 coord. query
         P2 =ObjectGet(His_Name,OBJPROP_PRICE2);// p2 coord. query
         Col=ObjectGet(His_Name,OBJPROP_COLOR); // Color query
         if(T1!=T_Bar || T2!=T_Bar || // Incorrect coord. or color:
            (Line==1 && (P1!=High[Ind_Bar] || P2!=  Low[Ind_Bar])) ||
            (Line==2 && (P1!=Open[Ind_Bar] || P2!=Close[Ind_Bar])) ||
            (Open[Ind_Bar] Close[Ind_Bar] && Col!=Red)  ||
            (Open[Ind_Bar]==Close[Ind_Bar] && Col!=Green)  )
           {
            ObjectDelete(His_Name);    // Delete object
            Create(Ind_Bar,Line);      // Create correct object
           }
//--------------------------------------------------------------- 8 --
        }
     }
   WindowRedraw();                     // Image redrawing
   return;                             // Exit start()
  }
//--------------------------------------------------------------- 9 --
int deinit()                           // Spec. function deinit()
  {
   string Name_Del[1];                 // Array declaring
   int Quant_Del=0;                    // Number of objects to be deleted
   int Quant_Objects=ObjectsTotal();   // Total number of all objects
   ArrayResize(Name_Del,Quant_Objects);// Necessary array size
   for(int k=0; k<=Quant_Del; i++)     // Delete objects with names.. 
      ObjectDelete(Name_Del[i]);       // .. that array contains
   return;                             // Exit deinit()
  }
//-------------------------------------------------------------- 10 --
int Create(int Ind_Bar, int Line)      // User-defined function..
  {                                    // ..of objects creation
   color Color;                        // Object color
   datetime T_Bar=Time [Ind_Bar];      // Bar opening time
   double   O_Bar=Open [Ind_Bar];      // Bar open price
   double   C_Bar=Close[Ind_Bar];      // Bar close price
   double   H_Bar=High [Ind_Bar];      // Bar maximum price
   double   L_Bar=Low  [Ind_Bar];      // Bar minimum price
 
   string Nom_Lin =Line + "_";         // String - line number
   // string Nom_Lin  = DoubleToStr(Line,0)+"_";// Can be so
   string Str_Time=TimeToStr(T_Bar);   // String - open time.     
   string His_Name=Prefix+Nom_Lin+Str_Time;// Name of created object
   if (O_Bar < C_Bar) Color=Blue;      // Choosing the color depending on..
   if (O_Bar >C_Bar) Color=Red;        // .. parameters of the bar
   if (O_Bar ==C_Bar) Color=Green;
 
   switch(Line)                        // Thin or thick line
     {
      case 1:                          // Thin line
         ObjectCreate(His_Name,OBJ_TREND,0,T_Bar,H_Bar,T_Bar,L_Bar);
         break;                        // Exit from switch
      case 2:                          // Thick line
         ObjectCreate(His_Name,OBJ_TREND,0,T_Bar,O_Bar,T_Bar,C_Bar);
         ObjectSet(   His_Name, OBJPROP_WIDTH, 3);// Style     
     }
   ObjectSet(    His_Name, OBJPROP_COLOR, Color); // Color
   ObjectSet(    His_Name, OBJPROP_RAY,   false); // Ray
   ObjectSetText(His_Name,"Object is created by the EA",10);// Description
   return;                             // Exit user-defined function
  }
//-------------------------------------------------------------- 11 --
In order to create graphical objects, the user-defined function Create() (blocks 10-11) is used in the program. The variable Ind_Bar that indicates the index of bar the object should be created on, and Line, the object number (line 1 or 2), are used as the assignable parameters in this function.

Three components are used when forming the name of the object to be created:

   string His_Name = Prefix+Nom_Lin+Str_Time;// Name of created object
The value of the Prefix variable is specified by the programmer in the head part of the program and it is not changed during the program execution:

string     Prefix     = "Paint_";
The value of the Nom_Lin variable is obtained as a result of calculations:

   string Nom_Lin  = Line + "_";       // String - line number
// string Nom_Lin  = DoubleToStr(Line,0)+"_";// Can be so
Here the value of the integer variable (during calculation in the first part of expression) is transformed into the type of the higher priority, namely, into the string type. As a result, the Nom_Lin receives "1_" or "2_" values depending on the value of the Line variable.

In order to calculate the value of the Str_Time variable the TimeToStr() function of data transformation is used:

   string Str_Time = TimeToStr(T_Bar); // String - open time
Please note that the TimeToStr() function has default values. In this case, these are these values that are necessary: "yyyy.mm.dd hh:mi"; there is no need to use seconds additionally, because the minimum timeframe is equal to 1 minute.

We could also apply the following method of Str_Time calculation to be used in the object name:

   string Str_Time = T_Bar;
In this case, the Str_Time would obtain a value equal to the number of seconds lapsed since 01.01.1970. In order to see the difference, we can develop a program that contains the following code:

int init()
  {
   string String_Time = TimeToStr(Time[0]); // Time in the format
   string String_Sec  = Time[0];            // Number of seconds
   Alert("String_Time = ",String_Time,"  String_Sec = ",String_Sec);
   return;
  }
The following message (according to the time of zero bar opening) will be displayed on the screen as a result of the program execution:

	String_Time = 2007.03.22 19:10 String_Sec = 1174590600


The first alternative that is implemented in the strings.mq4 EA is a bit more informative, so the preference is given to it, in this case (the alternatives are equivalent in terms of composing an algorithm).

The object named His_Name is created in the subsequent lines of the user-defined function Create(). It contains the information about the bar opening time with the parameters that correspond to the number of the "Line" line and also the color depending on bar characteristics. The value of the text description is specified for every object, "Object is created by EA", as well.

The Create() function is called in the program from two places: from the special function init() for the initial creation of objects, and from the special function start() to re-create the object, if necessary, in case it was deleted or modified by the user. The object names in start() (blocks 4-6) are formed in the same way as in other parts of the program.

The first coordinate of the considered object is defined in block 6-7. If the object is not found at this time, it will be created by the Create() function. And if the object exists, its other coordinates will be determined and the matching of its parameters with the bar properties will be checked (block 7-8). The object will be deleted and re-created (with the same name) with the correct properties, if any mismatch is detected.

Another problem is solved during the execution of the deinit() function: it is necessary to delete only the objects that have been created by the program from the aggregate of all objects in the symbol window. This is performed in two stages: at the first stage, the names of all objects that should be deleted are memorized to the Name_Del[] array, and then they will be deleted in an individual cycle. The total number of objects in the window (including those created by the program and placed manually by the user) is calculated using the ObjectsTotal() function:

   int Quant_Objects=ObjectsTotal();   // Total number of ALL objects
The number of bars to be colored is set by the user in an external variable, i.e., it is unknown in advance how many objects should be deleted. So the string array that contains the names of the objects to be deleted is declared with the number of elements equal to 1. Further, its size is programmatically changed - the number of elements is increased to the total number of objects.

   ArrayResize(Name_Del,Quant_Objects);// Necessary array size
In order to select the objects that have been created by the EA, the deinit() function contains the cycle 'for' that analyzes the names of all objects.

      string Obj_Name = ObjectName(k); // Querying name of the object
The attribute that differentiates "our" objects from all the others is the "Paint_" prefix, with which the name of each program-created object starts. To analyze an object name, we should extract the first part (in this case, 6 symbols) from the string variable being the unique name of the object; then we should compare this value with that of the Prefix variable. If they match, this object should be deleted. If not, it should no be deleted.

StringSubstr() Function

string StringSubstr(string text, int start, int length=0)
The function extracts the substring from the text line starting from the specified position. The function returns the copy of the extracted substring. Otherwise, an empty string is returned. 

Parameters:

text - the line the substring should be extracted from;

start - the initial position of the substring. It can range from 0 to StringLen(text)-1;

length - the length of the substring to be extracted. If the value of this parameter is less than or equal to 0 or it is not even specified then the substring will be extracted from the specified position till the end of the line.

In the considered example, the substring is extracted from the object name as follows:

      string Head=StringSubstr(Obj_Name,0,6);// Extract first 6 symbols
In this case, the first 6 symbols are extracted from the Obj_Name string variable starting with the zero one. Please note that the count of all indexes (bars, arrays), entries in the orders list and also the number of the position in the line starts with 0, whereas the quantified count starts with 1.

The extracted substring (a string value) is assigned to the string variable Head. If the object name (and in the object itself) is created by the considered EA, the value of the extracted substring will be "Paint_". If another name is analyzed, then the desired value will be different. For example, the value of the extracted substring from the "StdDev Channel 23109" object name will be the following: "StdDev", and for the object named "Fibo 22800" it will be "Fibo 2".

In the subsequent lines, the value of the variable Head is compared to that the variable Prefix:

      if (Head == Prefix)              // The object beginning..
         {                             // .. with Paint_ is found
If these values are equal to each other, then the analyzed name of the object will be placed to the array Name_Del[] for the names of objects to be deleted. In the next "for" cycle, all the objects, the names of which are contained by the array, will be deleted (it should be noted separately that it is impossible to delete all the objects during the first cycle, because, in this case, the total number of objects and their numeration will be changed each time the object is deleted, which will result in the omission of some object names).

The price chart will have the following appearance during the execution of the strings.mq4 EA:

Рис. 142. Price chart colored using graphical objects (strings.mq4).

Besides the groups of objects that cover the price chart, two other objects placed manually by the user are displayed, as shown in Fig. 142; they are regression channel and Fibo levels. The objects created by the EA will be deleted, as soon as its execution is finished, and the objects created by the user will remain in the symbol window. This result is obtained due to the use of string functions in the program. They allow to create and analyze string values, including graphical object names.

String Functions

	Function
	Short Description

	StringConcatenate
	It forms a string from the given parameters and returns it. The parameters can be of any type. The number of parameters cannot exceed 64.

	StringFind
	Substring searching. It returns the number of the position in the line the desired substring starts with, or -1, in case the substring is not found.

	StringGetChar
	It returns the value of the symbol that is located at the specified position of the line.

	StringLen
	It returns the number of symbols in the line.

	StringSetChar
	It returns the copy of the line with the modified value of the symbol at the specified position.

	StringSubstr
	It extracts the substring that starts at the specified position in the text line. The function returns the copy of the extracted substring, if possible. Otherwise, an empty string is returned.

	StringTrimLeft
	The function cuts the carriage return characters, spaces and tabulation symbols from the left part of the string. The function returns the copy of the modified string, if possible. Otherwise, an empty string is returned.

	StringTrimRight
	The function cuts the carriage return characters, spaces and tabulation symbols from the right part of the string. The function returns the copy of the modified string, if possible. Otherwise, an empty string is returned.


Data Transformation Functions

	Function
	Summary Info

	CharToStr
	Transformation of the symbol code into a single-symbol string.

	DoubleToStr
	Transformation of the numeric value into a text string that contains the symbolic representation of the number with the specified accuracy format.

	NormalizeDouble
	Rounding off the number with the floating point to the specified accuracy. The calculated StopLoss, TakeProfit and also the open prcie of pending orders values must be normalized according to the accuracy that is stored in the defined Digits variable.

	StrToDouble
	Transformation of the string that contains the symbolic representation of the number into the number of "double" type (double-accuracy format with the floating point).

	StrToInteger
	Transformation of the string that contains the symbolic representation into the number of the "int" type (integer).

	StrToTime
	Transformation of the string that contains time and/or date in the "yyyy.mm.dd [hh:mi]" format into the number of the "datetime" type (number of seconds passed since 01.01.1970).

	TimeToStr
	Transformation of the value that contains the time expressed in seconds passed since 01.01.1970 into the string of the "yyyy.mm.dd hh:mi" format.


To get the detailed information about these and other functions take a look at the Documentation at MQL4.community, at MetaQuotes Software Corp. website or at the "Help" section of MetaEditor.

Date and Time

The online trading system MetaTrader 4 uses the indications of two time sources - the local (PC) time and the server time.

Local time - the time that is set on the local PC.

Server time - the time that is set on the server.

TimeLocal() Function

datetime TimeLocal()
The function returns the local PC time expressed in the number of seconds lapsed since 00:00 of the 1st of January 1970. Note: At testing, the local time is modeled and coincides with the modeled last-known server time.

A large majority of events that take place in the client terminal are considered with accordance to the server time. The time of tick coming, new bar beginning, order opening and closing is considered with accordance to the server time. To get the value of the server time that corresponds with the current time, the TimeCurrent() function should be used:

TimeCurrent() Function

datetime TimeCurrent()
The function returns the last known value of the server time (the time of the last quote coming) expressed in seconds lapsed since 00:00 of the 1st of January 1970. The client terminal updates the time of the last quote coming (together with other environment variables) before launching special functions for execution. Each tick is characterized with its own value of the server time that can be obtained using the TimeCurrent() function. During the execution, this value can only be changed as a result of the RefreshRates() function call and only if the information has been updated since the last execution of the RefreshRates() function, i.e., in case the new values of some environment variables have come from the server.

The time of bar opening, Time[i], does not coincide with the time of new tick coming, as a rule. The time of any timeframe bar opening is always divisible by the timeframe. Any first tick appeared within a timeframe is bar-forming; if there is no tick receipt within a timeframe, the bar will not be formed within the timeframe.

For example, the tick coming to the terminal at time (server) t0 results in forming a bar with the time opening equal to Time[i+2] (Fig. 143). The moment specified as the beginning of the timeframe does not concur with moment t0, though it can accidentally concur with it, in general. The subsequent ticks that come to the terminal within the same timeframe (at the moments of t1 and t2) can change the parameters of the bar, for example, maximum price or open price, but they do not affect the time of bar opening. The bar closing time is not considered in the online trading system MetaTrader 4 (formally, the time of the last tick coming within a timeframe or the beginning time of the next timeframe can be considered as the bar closing time, as shown in Fig. 143).

Fig. 143. Bar forming sequence in the online trading platform MetaTrader 4.

It is shown in Fig. 143 that it is possible that bars are not formed at some time periods that are equal to the timeframe. Thus, between time t5 of the tick coming and t6 of the next tick coming, the full timeframe is packed, so the new bar hasn't been formed at that time period. In this manner, the time of bar opening may differ from the time of opening of an adjacent bar by more than a whole timeframe, but it is always divisible by a timeframe. To demonstrate the sequence of bar forming, we can use the EA timebars.mq4 that outputs the time of tick coming and the time of bar opening:

//--------------------------------------------------------------------
// timebars.mq4
// The program is intended to be used as an example in MQL4 Tutorial.
//--------------------------------------------------------------------
int start()                            // Spec. function start()
  {
   Alert("TimeCurrent=",TimeToStr(TimeCurrent(),TIME_SECONDS),
         " Time[0]=",TimeToStr(Time[0],TIME_SECONDS));
   return;                             // Exit start()
  }
//--------------------------------------------------------------------
The results of the EA timebars.mq4 working are shown in Fig. 144. It is obvious that the first tick at the regular time period of 1 minute duration came at 14:29:12, at the same time a new bar was formed with the opening time - 14:29:00. Please note that the right column of the message box displays the server time, the left column displays the local time.

Fig. 144. Bar forming sequence in the online trading system MetaTrader 4.

In case the ticks come rarely (for example, the period between the end of the European session and the beginning of the Asian session), you can observe another phenomenon during the execution of timebars.mq4: the opening time of the adjacent bars can differ from each other by more than 1 minute (for one-minute timeframe). At the same time, the indexing of bars is saved in succession, without spaces.

The server time of servers in different dealing centers may vary. The time of beginning and finishing trades is set on each server individually and it can disagree with the begining and the end of the regular day. Some dealing centers, for example, have the settings that perform trade opening on Sunday at 23:00 of server time. This results in forming of incomplete daily bars, their practical duration is equal to one hour (Fig. 145).

Fig. 145. Different bar history in different dealing centers.

The usage of date and time functions is rather easy in MQL4. Some of them transform the server and the local time in seconds lapsed since 00:00 of the 1st of January 1970 into an integer number that corresponds with an hour, a day, etc. Other functions return an integer number that corresponds with the current hour, day, minute, etc.

TimeSeconds (), TimeMinute(), TimeHour(), TimeDay(), TimeMonth(), TimeYear(), TimeDayOfWeek () and TimeDayOfYear() Functions

This is a group of functions that return the number of seconds lapsed from the beginning of the minute, or minute, hour, day, month, year, day of week and day of year for the specified time. For example:

int TimeMinute(datetime time)
The function returns minutes for the specified time.

Parameters:

time - the date expressed in number of seconds that lapsed since 00:00 of the 1st of January 1970.

int TimeDayOfWeek(datetime time)
This function returns the day of week (0-Sunday,1,2,3,4,5,6) for the specified date.

Parameters:

time - the date expressed in number of seconds that lapsed since 00:00 of the 1st of January 1970.

The considered functions can be used for analysis of any bar opening time, for example. The EA named bigbars.mq4 intended for finding bars of a size that is not less than specified size is shown below.

//--------------------------------------------------------------------
// bigbars.mq4
// The code should be used for educational purpose only.
//--------------------------------------------------------------- 1 --
extern int Quant_Pt=20;                // Number of points
//--------------------------------------------------------------- 2 --
int start()                            // Spec. function start()
  {
   int H_L=0;                          // Height of the bar
   for(int i=0; H_L<Quant_Pt; i++)     // Cycle for bars
     {
      H_L=MathAbs(High[i]-Low[i])/Point;//Height of the bar
      if (H_L>=Quant_Pt)               // if the high bar is not found
        {
         int YY=TimeYear(  Time[i]);   // Year
         int MN=TimeMonth( Time[i]);   // Month         
         int DD=TimeDay(   Time[i]);   // Day
         int HH=TimeHour(  Time[i]);   // Hour         
         int MM=TimeMinute(Time[i]);   // Minute
         Comment("The last price movement more than ",Quant_Pt,//Message
         " pt happened ", DD,".",MN,".",YY," ",HH,":",MM);//output
        }
     }
   return;                             // Exit start()
  }
//--------------------------------------------------------------- 3 --
The bigbars.mq4 EA searches the nearest bar whose height (difference between minimum and maximum) is more than or equal to the value specified in the external variable Quant_Pt. The date and time of the found bar are outputted to the window of financial instrument by the Comment() function.

Seconds (), Minute(), Hour(), Day(), TimeMonth(), TimeYear(), DayOfWeek () and DayOfYear() Functions

This is the group of functions that return the current second, minute, hour, day, month, year, day of week and day of year for the last known server time. The last known server time is the server time that corresponds with the moment of the program launch (launch of any special function by the client terminal). The server time is not changed during the execution of the special function.

int Hour()
It returns the current hour (0,1,2,..23) of the last known server time. Note that the last known server time is modeled during testing.

int DayOfYear()
It returns the current day of the year (1 is the 1st of January,..,365(6) is the 31st of December), i.e., the day of the year of the last known server time. Note that the last known server time is modeled during testing.

The EA timeevents.mq4 that performs some actions as soon as the specified time comes can be used as an example of usage of the above functions.

//--------------------------------------------------------------------
// timeevents.mq4
// The code should be used for educational purpose only.
//--------------------------------------------------------------- 1 --
extern double Time_Cls=16.10;          // Orders closing time
bool Flag_Time=false;                  // Flag, there are no messages yet
//--------------------------------------------------------------- 2 --
int start()                            // Spec. start function
  {
   int    Cur_Hour=Hour();             // Server time in hours
   double Cur_Min =Minute();           // Server time in minutes
   double Cur_time=Cur_Hour + Cur_Min100; // Current time
   Alert(Cur_time);
   if (Cur_time>=Time_Cls)             // If the time for the event has come
      Executor();                      //.. then perform devised actions
   return;                             // Exit from start()
  }
//--------------------------------------------------------------- 3 --
int Executor()                         // User-defined function
  {
   if (Flag_Time==false)               // If there are no messages yet
     {                                 // .. then report (1 time)
      Alert("Important news time. Close orders.");
      Flag_Time=true;                  // Now the message has already appeared
     }
   return;                             // Exit user-defined function
  }
//--------------------------------------------------------------- 4 --
The server time is calculated in hours and minutes during the execution of the special function start() (blocks 2-3). The line:

   double Cur_time = Cur_Hour + Cur_Min100; // Current time
represents the current server time expressed as the real variable Cur_time. The use of real variables is convenient in comparison operations:

   if (Cur_time >= Time_Cls)           // If the time for the event has come
If the current time is more than or equal to the value of Time_Cls specified by user, then the Executor() user-defined function will be called for execution. In this example, the user-defined function puts out a message with trading recommendations. In general, this function may contain any code, for example, make trades, send e-mails, create graphical objects, etc.

Functions of Date and Time

	Function
	Description

	Day
	It returns the current day of month, i.e. the day of month of the last known server time. 

	DayOfWeek
	It returns the index number of the day of week (sunday-0,1,2,3,4,5,6) of the last known server time.

	DayOfYear
	It returns the current day of year (1 is the 1st of January,..,365(6) is the 31st of December), i.e. the day of year of the last known server time.

	Hour
	It returns the current hour (0,1,2,..23) of the last known server time at the moment of program start (the value is not changed during the execution of the program). 

	Minute
	It returns the current minute (0,1,2,..59) of the last known server time at the moment of program start (the value is not changed during the execution of the program). 

	Month
	It returns the number of the current month (1-January,2,3,4,5,6,7,8,9,10,11,12), i.e. the number of the month of the last known server time.

	Seconds
	It returns the number of seconds lapsed since the beginning of the current minute of the last known server time at the moment of program start (the value is not changed during the execution of the program).

	TimeCurrent
	It returns the last known server time (the time of last quote coming) expressed in the number of seconds that passed since the 00:00 January 1-st of 1970.

	TimeDay
	It returns the day of month (1 - 31) for the specified date.

	TimeDayOfWeek
	It returns the day of week (0-Sunday,1,2,3,4,5,6) for the specified date.

	TimeDayOfYear
	It returns the day (1 is the 1st of January,..,365(6) is the 31st of December) of year for the specified date.

	TimeHour
	It returns the hour for the specified time.

	TimeLocal
	It returns the local PC time expressed in the number of seconds lapsed since 00:00 of the 1st of January 1970.

	TimeMinute
	It returns minutes for the specified time.

	TimeMonth
	It returns the number of the month for the specified time (1-January,2,3,4,5,6,7,8,9,10,11,12). 

	TimeSeconds
	It returns the number of seconds passed from the beginnig of the specified time.

	TimeYear
	It returns the year for the specified date. The returned value can be within the range of 1970-2037.

	Year
	It returns the current year, i.e. the year of the last known server time.


To get the detailed information about these and other functions, refer to the Documentation at MQL4.community, at MetaQuotes Software Corp. website or at the "Help" section of MetaEditor.

File Operations

In MQL4, it is possible to work with files containing a certain set of information. It may become necessary to write information in a file or to read it from a file for several reasons.

A file can be used to deliver information to another program. In this case, the file can be created by an application program and used by it as an information receiver. For example, the trading history of an account can be written to a file at the execution of an application. This file can be later opened by another program (e.g., Excel for drawing a balance graph).

In other cases, there is a need to deliver some information, for example, the news timetable, to an application. An executable program (e.g., an Expert Advisor) can read this information from the previously prepared file and consider it during calculating for graphical displaying of the messages on the screen or for making of trade decisions.

File Names and Directories

The name of a working file must be composed according to the requirements of the operating system. The name of any file used in MQL4 consists of two parts: the file name and the file extension separated by a dot, for example, News.txt. Technically, a file name has no relation to the file content, so a file name and extension can be set voluntarily by the programmer. A file name is usually chosen so that it represents the information the file contains.

Most programs are automatically launched on the user's PC, if the file is double-clicked with the mouse button. According to the file extension, the operating environment loads one or another program displaying the file content. Therefore, you should assign the file extension considering the program (if necessary) that will usually be used to view the file.

The most popular file types (the type is determined by its extension) are the following:

- .txt - text file, for viewing you should use Notepad, Word, FrontPage, etc.;

- .csv - file for building tables in Excel;

- .htm - file to be viewed in a browser, i.e. Internet Explorer, Netscape Navigator, etc.

There are three folders (with subfolders) that can contain working files:

- Terminal_folder\Experts\History\current broker\ - for history files;

- Terminal_folder\Experts\Files\ - for common usage;

- Terminal_folder\Tester\ Files\ - for files the are used for testing.

A working file can be saved in one of these folders or in their subfolders. In case of no available folder at the moment of file saving, the folder will be automatically created by the client terminal. Working with files in other directories is not involved.

Modes of File Operations

The technology of interaction between an application and a working file has several modes. In general, a file can be opened by several programs at the same time (within a PC or several PCs connected to the network). At the same time, the operational environment provides the full access to the file, namely the right to read the file and write the information in it, only to one program. The other programs can only read it. For example, if My_text.doc has already been opened by a text editor, then all the other programs will receive the notification before opening the file:

Fig. 146. Dialog box that appears when a program tries to access to the file that has already been opened by another program.

The execution of this technology guaranties that a file won't be modified simultaneously by two different programs. In order to allow an applicable program to interact with a file, you should open that file first. The mode of opening a file is specified in the FileOpen() function.

An application program can open several working files at a time. In order to allow the program to differentiate one file from another the file descriptor is set in accordance to every opened file.

File descriptor – unique number of the file that is opened by the program at the moment.

The FileOpen() function will return some integer value (this value is usually assigned to the 'handle' variable), if a file is opened successfully. That value is the file descriptor. Most functions that are intended to work with files suppose the use of a file descriptor as one of the formal parameters.

Function FileOpen()

 int FileOpen(string filename, int mode, int delimiter=';')
The function opens a file for inputing and/or outputting. The function returns a file descriptor or -1, in case of failure. Files can only be opened in the Terminal_folder\Experts \Files\ folder or in the Terminal_folder\Tester\Files\folder (in case of EA testing) or in their subfolders.

Parameters:

filename - name of the file;

mode - the mode of file opening; it can have the following values (or their combinations): FILE_BIN, FILE_CSV, FILE_READ, FILE_WRITE;

delimiter - the separator sign for csv-files. It is ';' by default.

The FILE_READ mode of file opening implies that a file will be used only for being read by a program. A trial to open a file in this mode can fail, in case of no available file with the specified name.

The FILE_WRITE mode of file opening involves that a file will be used for writing in by a program. A try to open a file in this mode results in opening a file of a zero length. Even if there was some information in the file before opening, it will be erased. A try to open a file in this mode can fail, in case the file was previously opened by another program (in the writing mode).

It is allowed to open a file in the FILE_READ|FILE_WRITE mode. This mode involves the possibility of reading and writing to a file. This mode is used, if you need to add some information to the file that already contains some other information. The function implies the obligatory usage of one of the modes, FILE_READ or FILE_WRITE, or their combination.

The FILE_BIN mode of file opening defines processing a working file as a binary one. The FILE_CSV mode of file opening defines processing of a working file as a text one. The function involves obligatory usage of one of the FILE_BIN or FILE_CSV modes. The simultaneous usage of FILE_BIN and FILE_CSV modes is prohibited

The function requires obligatory combination of FILE_READ, FILE_WRITE or FILE_READ|FILE_WRITE modes with the FILE_BIN or FILE_CSV mode. For example: it is necessary to use the combination of FILE_CSV|FILE_READ to read the information from a text file, and it is necessary to use the FILE_BIN|FILE_READ|FILE_WRITE combination to add an entry to a binary file.

No more than 32 files can be opened simultaneously within an executable module (of an applicable program, e.g., an Expert Advisor). The descriptors of the files that are opened in the module cannot be passed to other modules (libraries).

Content of File Entries

The information entries are written to a file without spaces with any combination of modes. The information entries are added one by one when using the FILE_BIN mode for forming a file. Depending on the type of information that is written to a file (and the functions that are used to do it) the symbols combination representing the end of the line ("\r\n") can be written between the groups of entries. The information entries are separated by file separators (usually ';') when forming a file in the FILE_CSV mode, and the groups of entries (that compose a line) are separated with the combination of symbols that represent the end of the line ("\r\n").

File separator - special symbol; the entry that is written to a file to separate the information lines.

The file separator is used to separate the information entries only in the csv-files.

The common principle for entries composition in any files is that these entries are added according to the specific sequence without spaces. Properly, the entry consists of continuous sequence of symbols. Any file can be read by any program and (depending on the rules implemeted in it) can be displayed in some form on the screen. For example: we have the File_1.csv file that contains:

int FileOpen(string filename, int mode, int delimiter=';')
The File_1.csv file will be displayed in different ways in different text editors:

Fig. 147. File_1 representation in different programs (File_1.csv).

In this case, the "\r\n" symbol combination was interpreted by each of the programs (Excel and Notepad) as the evidence for formatting sequence: the sequence of symbols is represented in the next line after the "\r\n" combination of symbols, and the "\r\n" combination itself is not displayed in the editing window. At the same time, Excel is a table editor, so the ";" symbol was interpreted by the program as a separator of information to columns. Draw your attention that the ";" symbol is not displayed in the Excel window. Notepad is a text editor. The rules implemented in it do not suppose the division of information into columns, so the ";" symbol was not interpreted as a file separator, but it was interpreted as a part of information, so it is displayed on the screen.

The specified symbols (";" and "\r\n") are used to separate entries in MQL4.

Fig. 148. Variety of entries in working files.

The structure of information writing in different types of files is represented on the fig. 148. The top line shows the csv-file content, the bottom three lines show the structure of binary files. All these files are composed according to the rules of one or another function of writing in the file.

An entry in the csv-file is the sequence of string values (string type) that are separated with the file separator or with the sign of the end of the line. Both of them are interpreted as a sign of the end of the informative read value part when reading information (using standard MQL4 function for file reading). The string value can have the different length and it is unknown how much symbols are there, so the reading is performed before one of the separators is located.

The entries in two types of binary of binary files represent the sequences of data without any separators. This sequence of writing in is governed by the fixed length for a data of different types: 4 bytes for a data of the "int", "bool", "datetime" and "color" types, and 8 bytes (or 4 bytes, depending on the parameters of writing function) for a data of "double" type. In this case, there is no need of separators, because the reading is performed by the standard function for reading data of a specified type with a specified length. The last (the bottom one on the fig. 148) binary file contains the data of string type that is separated with the end of the line sign.

File pointer - a position in the file the reading of the next value starts from.

The "File pointer" notion is the same with "cursor" notion. The file pointer is defined with the position in the file. As far as reading goes on the pointer is moving to the right per one or several positions.

	Problem 36. Read the information about the important news from the file and display the graphical objects on the price chart (vertical lines) in accordance to the time of news publication.


Let the Terminal_Folder\Experts\Files\ folder contains the News.csv working file with the following content:

Fig. 149. Content of working file News.csv.

In this case, the file contains the information about five events that will happen in different countries at a different time. Each line contains two entries. The first entry is the string value that represent the information about date an time of the event. The second entry is the text description of the event. First three symbols of the second entry contain the name of currency (country) that the event concerns.

The solution consists of two parts. First of all we need to read the information from the working file and then use the received values as the coordinates of the graphical objects. The reading of information is performed by the FileReadString() function.

FileReadString() Function

 string FileReadString(int handle, int length=0)
The function reads the line from the current position of the file. It is suitable both for CSV and binary files. The line will be read till the separator is met in the text file. The specified number of symbols will be read in the binary files. In order to receive the information about an error you should call the GetLastError() function.

Parameters:

handle - the file descriptor that is returned by the FileOpen() function;

length - the number of characters to be read.

The need in news information processing appears only once at the beginning of trading, so, in this case, we can use a script to solve the problem 36. The timetablenews.mq4 script is intended to read the information from the file and display the graphical objects in the symbol window.

//--------------------------------------------------------------------
// timetablenews.mq4
// The code should be used for educational purpose only.
//--------------------------------------------------------------- 1 --
int start()                            // Spec. function start()
  {
//--------------------------------------------------------------- 2 --
   int Handle,                         // File descriptor
       Stl;                            // Style of vertical line
   string File_Name="News.csv",        // Name of the file
          Obj_Name,                    // Name of the object
          Instr,                       // Name of the currency
          One,Two,                     // 1st and 2nd name of the instr.
          Text,                        // Text of event description
          Str_DtTm;                    // Date and time of the event (line)
   datetime Dat_DtTm;                  // Date and time of the event (date)
   color Col;                          // Color of the vertical line
//--------------------------------------------------------------- 3 --
   Handle=FileOpen(File_Name,FILE_CSV|FILE_READ,";");// File opening
   if(Handle<0)                        // File opening fails
     {
      if(GetLastError()==4103)         // If the file does not exist,..
         Alert("No file named ",File_Name);//.. inform trader
      else                             // If any other error occurs..
         Alert("Error while opening file ",File_Name);//..this message
      PlaySound("Bzrrr.wav");          // Sound accompaniment
      return;                          // Exit start()      
     }
//--------------------------------------------------------------- 4 --
   while(FileIsEnding(Handle)==false)  // While the file pointer..
     {                                 // ..is not at the end of the file
      //--------------------------------------------------------- 5 --
      Str_DtTm =FileReadString(Handle);// Date and time of the event (date)
      Text     =FileReadString(Handle);// Text of event description
      if(FileIsEnding(Handle)==true)   // File pointer is at the end
         break;                        // Exit reading and drawing
      //--------------------------------------------------------- 6 --
      Dat_DtTm =StrToTime(Str_DtTm);   // Transformation of data type
      Instr    =StringSubstr(Text,0,3);// Extract first three symbols
      One=StringSubstr(Symbol(),0,3);// Extract first three symbols
      Two=StringSubstr(Symbol(),3,3);// Extract second three symbols
      Stl=STYLE_DOT;                   // For all - dotted line style
      Col=DarkOrange;                  // For all - this color
      if(Instr==One || Instr==Two)     // And for the events of our..
        {                              // .. symbol..
         Stl=STYLE_SOLID;              // .. this style..
         Col=Red;                      // .. and this color of the vert. line
        }
      //--------------------------------------------------------- 7 --
      Obj_Name="News_Line  "+Str_DtTm;              // Name of the object
      ObjectCreate(Obj_Name,OBJ_VLINE,0,Dat_DtTm,0);// Create object..
      ObjectSet(Obj_Name,OBJPROP_COLOR, Col);       // ..and its color,..
      ObjectSet(Obj_Name,OBJPROP_STYLE, Stl);       // ..and style..
      ObjectSetText(Obj_Name,Text,10);              // ..and description
     }
//--------------------------------------------------------------- 8 --
   FileClose( Handle );                // Close file
   PlaySound("bulk.wav");              // Sound accompaniment
   WindowRedraw();                     // Redraw object
   return;                             // Exit start()
  }
//--------------------------------------------------------------- 9 --
The used variables are opened and described in block 2-3 of the EA. An attempt to open the file and the analysis of the results of this operation are performed in block 3-4. The FileOpen() function is used to open the file:

   Handle=FileOpen(File_Name,FILE_CSV|FILE_READ,";");// File opening
An attempt to open the file is not always successful. It can fail, if the file with the specified name is not available. When file opening fails (the file descriptor is a negative number) the necessary text message is displayed to the user and the execution of the start() function stops.

In case of successful opening of a file, control is passed to the operator of the "while" cycle (blocks 4-8). The reading of data from the file (block 5-6), transformation of data and its analysis (6-7 blocks) and creation of the graphical object with coordinates and parameters corresponding the last read information (block 7-8) are performed at each iteration.

The execution of the "while" cycle continues until the file pointer reaches the end of the file, i.e., there will be no information remaining to the right of the pointer. The FileIsEnding() function is used to analyze the position of the file pointer.

FileIsEnding() Function

bool FileIsEnding(int handle)
The function returns TRUE if the file pointer is at the end of the file, otherwise it returns FALSE. In order to receive the information about an error you should use the GetLastError() function. The GetLastError() function will return the ERR_END_OF_FILE (4099) error, in case the end of the file is reached during the reading.

Parameters:

handle - file descriptor that is returned by the FileOpen() function.

The represented solution (timetablenews.mq4) involves that any number of news can be written to the News.csv file. News.csv file contains five entries corresponding to five events (news) in the mentioned example (fig. 149). In general, the number of lines may be from 0 to 20-30, depending on the amount of real events that must take place this day.

Reading entries from the file (that is identified by the "handle" variable) is performed in blocks 5-6:

      Str_DtTm =FileReadString(Handle);// Date and time of the event (date)
      Text     =FileReadString(Handle);// Text of event description
      if(FileIsEnding(Handle)==true)   // File pointer is at the end
         break;                        // Exit reading and drawing
The first and second lines of block 5-6 perform reading the information from the file until the nearest separator is met. The third and fourth lines perform checking: is the file pointer at the end of the line. If not, then the graphical objects will be formed via two read values further in the cycle. If it was initially known about the number of entries, then the analysis that is performed in the third and fourth lines would not be necessary. In this case, we would hardly specify the number of iterations in the cycle (e.g. 5) and would not perform an extra checking.

However, the number of entries is unknown, in this case. At the same time, in this example every event is described with two values that compose a line of the following kind: value, file separator, value, end of the line sign. In this case, it is supposed that if there is an entry (first value in the line) then the other one exists; but if there is no first entry then the second one does not exist, so there is no event and there is no need to create a graphical object. If both entries or one of them does not exist the pointer will move to the end of the file (i.e. the position in the file where no data to the right of the pointer exist) when a try to read it is performed. The checking performed in block 3-4 allows to discover this fact. If the noted checking (last two lines in block 5-6) is deleted, then needless object will be created while the program is running. Only after that the condition of "while" cycle ending will trigger and the control will be passed to block 8-9. In general, you should consider the logic of data representation in the file, sequence order of entries and separators, the number of lines, etc. while composing an algorithm for file reading. Each certain circumstance requires an individual algorithm.

Data read from the file has the string type. In order to use the received values for creating graphical objects you should transform the data to the necessary type. In block 6-7, the first (read in the next line) value is transformed to the "datetime" value and further will be used as the coordinate of the graphical object that corresponds the event. First three symbols from the second read string value are compared with the first and second triplet of symbols in the symbol name. If there is a coincidence then the graphical object receives the corresponding parameters: line style - solid and color - red (block 7-8). In other cases, the objects are displayed with the orange dotted line. You can observe the news lines in the symbol window as the result of the script execution:

Fig. 150. Graphical objects in the symbol window after timetablenews.mq4 execution.

In such a way, the script can be executed in any symbol window. At the same time, every window will contain the solid red line that represent the events that concern this specific symbol, and the dotted lines that represent the vents concerning the other symbols' events. To display the text descriptions of the objects you should check the "Show object description" option in the Properties of security window (F8) => Common.

The previously opened file is closed in block 8-9 after the problem is solved, namely all the necessary objects are created. The file should be closed for the following reasons: on the first hand - not to spare extra PC resources and on the second hand to allow the other programs to access the file in the writing mode. It should be considered as normal to close the file as soon as all the information is read from it (or written in it) and its usage is not necessary anymore. The closing of file is performed by the FileClose() function.

FileClose() Function

 void FileClose(int handle)
The function performs closing of a file that was previously opened by the FileOpen() function.

Parameters:

handle - file descriptor that is returned by the FileOpen() function.

In order to allow the trader to practically use the timetablenews.mq4 script, it must keep the method for creation of a file that contains the news timetable of some period. This type of file can be created using any text editor, however, in this case, the possibility of an error remains (sometimes a separator can be not specified erroneously). Lets examine a variant of working file creation using MQL4. 

	Problem 37. Represent the code of the EA that creates a file for news timetable.


In general, an EA can be destined for creation of a file that contains any number of news. The examined here createfile.mq4 EA creates the working file that contains the information about not more than five events.

//--------------------------------------------------------------------
// createfile.mq4
// The code should be used for educational purpose only.
//--------------------------------------------------------------- 1 --
extern string Date_1="";   // 2007.05.11 10:30
extern string Text_1="";   // CHF Construction licenses
extern string Date_2="";   // 2007.05.11 12:00
extern string Text_2="";   // GBP Refinance rate,2%,2.5%
extern string Date_3="";   // 2007.05.11 13:15
extern string Text_3="";   // EUR Meeting of G10 banks governors
extern string Date_4="";   // 2007.05.11 15:30
extern string Text_4="";   // USD USA unemployment rate
extern string Date_5="";   // 2007.05.11 18:30
extern string Text_5="";   // JPY Industrial production
//--------------------------------------------------------------- 2 --
int start()                            // Spec. function start()
  {
//--------------------------------------------------------------- 3 --
   int Handle,                         // File descriptor
   Qnt_Symb;                           // Number of recorded symbols
   string File_Name="News.csv";        // File name
   string Erray[5,2];                  // Array for 5 news
//--------------------------------------------------------------- 4 --
   Erray[0,0]=Date_1;                  // Fill the array with values
   Erray[0,1]=Text_1;
   Erray[1,0]=Date_2;
   Erray[1,1]=Text_2;
   Erray[2,0]=Date_3;
   Erray[2,1]=Text_3;
   Erray[3,0]=Date_4;
   Erray[3,1]=Text_4;
   Erray[4,0]=Date_5;
   Erray[4,1]=Text_5;
//--------------------------------------------------------------- 5 --
   Handle=FileOpen(File_Name,FILE_CSV|FILE_WRITE,";");//File opening
   if(Handle==-1)                      // File opening fails
     {
      Alert("An error while opening the file. ",// Error message
              "May be the file is busy by the other applictiom");
      PlaySound("Bzrrr.wav");          // Sound accompaniment
      return;                          // Exir start()      
     }
//--------------------------------------------------------------- 6 --
   for(int i=0; i<=4; i++)             // Cycle throughout the array
     {
      if(StringLen(Erray[i,0])== 0  || // If the value of the first or..
         StringLen(Erray[i,1])== 0)    // ..second variable is not entered
         break;                        // .. then exit the cycle
      Qnt_Symb=FileWrite(Handle,Erray[i,0],Erray[i,1]);//Writing to the file
      if(Qnt_Symb < 0)                 // If failed
        {
         Alert("Error writing to the file",GetLastError());// Message
         PlaySound("Bzrrr.wav");       // Sound accompaniment
         FileClose( Handle );          // File closing
         return;                       // Exit start()      
        }
     }
//--------------------------------------------------------------- 7 --
   FileClose( Handle );                // File closing
   Alert("The ",File_Name," file created.");// Message
   PlaySound("Bulk.wav");              // Sound accompaniment
   return;                             // Exit start()
  }
//--------------------------------------------------------------- 8 --
The initial information is entered to the program using the external variables of the "string" type (block 1-2). The variables are opened and described in block 3-4. To make the processing convenient the data is written to the Erray[][] string array. Every event (information that characterize news) is represented by two elements of the array in the second dimension. The size of the first dimension (the number of lines in the array) is defined with the number of news, in this case, 5. In order to prevent the manual entering of values while trying the EA on a demo-account you can load the settings of the EA file example_news.set; the file of the EA setting should be located in the Terminal_folder\presets \ folder.

Block 5-6 performs file opening. If the operation failed then the start() function ends working after the user has received the message. If the file is opened successfully then the control will be passed to the "for" cycle operator in block 6-7. In general, the number of input values, the size of the Erray array and the number of iterations can be increased to the necessary quantity.

The checking is performed every iteration: is one of the entered values empty. The length of the Erray array values is calculated for this aim. If one of them has the zero length then it is considered as the absence of the current and the next events, so the current iteration interrupts. The writing of values of two elements of the array to the file goes on as far as the empty value of the element is found. The FileWrite() function is used for writing the values to the csv-file.

FileWrite() Function

 int FileWrite(int handle, ...)
The function is intended for writing the information to a csv-file, the separator between the information is included in automatically. The sign representing the end of the line "\r\n" is added to the file after the information writing. The information is transformed from the numeric to the text format when outputted (see Print() function). The function returns the number of written symbols or the negative value, in case an error occur.

Parameters:

handle - file descriptor that is returned by the FileOpen() function;

... - data separated with commas. It cannot be more than 63 parameters.

The data of the "double", "int" types is automatically transformed to the string (the data of "color", "datetime" and "bool" types is considered as the integer numbers of the "int" type and transformed to the string, as well), the data of the "string" type is output as is, without transformation. The arrays cannot be passed as the parameters; arrays must be entered elementwise.

In the considered example the information is written to the file in the following line:

      Qnt_Symb=FileWrite(Handle,Erray[i,0],Erray[i,1]);//Writing to the file
The separator (the symbol that is used as a separator is specified in the file opening function FileOpen(), in this case, ';') will be written after the Erray[i,0] value when writing to the file. The sign representing the end of the line "\r\n" is automatically placed at the end of the the FileWrite() function execution, i.e. at the end of writing. The same entry will be written on each next iteration of the "for" cycle. Every new entry starts from the position where the file separator of the last writing is placed. At the same time, the values of the next elements of the 'Erray' will be written to the file (indexes of the elements are increased by 1 on every iteration).

If the current writing to the file is successful the the control is passed to the next iteration. If the writing in the file fails then the file will be closed by the FileClose() function after the message is displayed to the user, and the start() function finishes its working. If all writings to the file are successfully performed then the control is passed to the file closing function FileClose() in block 7-8 after the execution of the "for" cycle is finished. In this case, the message about the successful file creation is displayed, after that the start() function execution will be finished. The News.csv file shown on the fig. 149 will be created after the EA execution is finished.

Function for Performing File Operations

	Function
	Summary Info

	FileClose
	The closing of the file that was previously opened by the FileOpen() function.

	FileDelete
	Deleting of the file. The files can only be deleted if they are located at the terminal_folder\experts\files (terminal_folder\tester\files, in case of testing the EA) folder or in its subfolders.

	FileFlush
	Flushing all the information that is left in the file input-output bufer to the hard disk.

	FileIsEnding
	Returns TRUE if the file pointer is at the end of the file, otherwise - returns FALSE. If the end of the file is reached during the file reading, the GetLastError() function will return the ERR_END_OF_FILE (4099) error.

	FileIsLineEnding
	It returns TRUE, if the file pointer is at the end of the line of the CSV-file. Otherwise, it returns FALSE.

	FileOpen
	Opens a file for inputting and/or outputting. The function returns the file descriptor of the opened file of -1, in case it fails.

	FileOpenHistory
	Opens a file in the current history folder (termial_folder\history\server_name) or in its subfolders. The function returns the file descriptor or -1, in case it fails.

	FileReadArray
	The function reads the specified number of elements from the binary file to the array. The array must have enough size before reading. The function returns the number of practically read elements.

	FileReadDouble
	The function reads the number of double accuracy with the floating point (double) from the current position of the binary file. The size of the number may the following: 8 bytes (double) and 4 bytes (float). 

	FileReadInteger
	The function reads the integer number from the current position of the binary file. The size of the number may be the following: 1, 2 or 4 bytes. If the size of the number is not specified then the system will try to read it as it was the 4 byte integer number.

	FileReadNumber
	Reading the number from the current position of the CSV-file until the separator is met. It can be applied only to csv-files.

	FileReadString
	The function reads the line from the current position of the file. It can be applied both for csv and binary files. The line in the text file will be read until the separator is met. The specified number of symbols in the line will be read in the binary files.

	FileSeek
	The function moves the separator to the new position that is the displacement from the beginning, end or the current position of the file in bytes. The next reading or writing starts from the new position. If the pointer moving is performed successfully then the function will return TRUE, otherwise - FALSE.

	FileSize
	The function returns the size of the file in bytes.

	FileTell
	The function returns the shift of file pointer from the beginning of the file. 

	FileWrite
	The function is intended to write the information to the csv-file, the separator is placed automatically between the information. The end of the line sign "\r\n" is added to the file after the writing is finished. The numeric data is transformed to the text format during the ouptputting process. The function returns the the number of written symbols or a negative value if an error occurs.

	FileWriteArray
	The function writes the array to the binary file.

	FileWriteDouble
	The function writes the number with the floating point to the binary file.

	FileWriteInteger
	The function writes the integer number value in the binary file.

	FileWriteString
	The function writes the line to the binary file from the current position. It returns the number of practically written bytes or a negative value, in case an error occurs.


To get the detailed information about these and other functions you should take a look at the documentation at MQL4.community, at MetaQuotes Software Corp. website or at the "Help" section of MetaEditor.

